```
// This ImageJ macro reassigns the background colors to (0,0,0) for all tif images
// within a directory. The background must be a low-saturation color, such
// as gray, the foreground images must be relatively high in saturation, and light
// conditions must be uniform. A dialog box appears to ask
// the location of the directory of apple images. The macro requires the plugins
// "Color Space Converter" and "Image Calculator Plus," both of which are available
// at http://rsbweb.nih.gov/ij/plugins/index.html. To change the threshold values
// for apple isolation from the bacground, simply change the values in the
// changeValues() function below. The default threshhold is a saturation level of
// 0.40/1.0. All values below 0.40 are assigned a value of (0,0,0), while all
// values above 0.40 retain their original values. The macro was developed to
// isolate apple images (high-saturation) from a neutral (low-saturation)
// background but could be used for any image isolation with similar
// requirements.


macro "Apple Image Isolator" {
        dir = getDirectory("Choose a source directory that contains apple images");
        list = getFileList(dir);
        setBatchMode(true);
        for (i=0; i<list.length; i++) {
                path = dir+list[i];
                open(path);
                rename("img1");

                run("Stack to RGB");
                run("Color Space Converter", "from=RGB to=HSB white=D65 separate");
                close();
                changeValues(0, 0.40, 0);
                changeValues(0.1,1,1);
                rename("img2");
                imageCalculator("Multiply create stack", "img1","img2");

                dotIndex = lastIndexOf(path, ".");
                if (dotIndex!=-1)
                path = substring(path, 0, dotIndex); // remove extension to create
                //uncluttered file name
                saveAs("Tiff", path+"clipped");
                close();
                close();
                close();
                close();
                close();
                }
        }
```